Постановка задачи теории упругости для чистого изгиба бруса

Полная система уравнений и граничных условий, описывающая напряженно-деформированное состояние системы, в рамках линейной теории упругости задается формулами (6-11):

Уравнения равновесия в трехмерном случае имеют вид

$$\begin{cases} \frac{\partial \sigma_{xx}}{\partial x} + \frac{\partial \sigma_{xy}}{\partial y} + \frac{\partial \sigma_{xz}}{\partial z} = 0 \\ \frac{\partial \sigma_{yx}}{\partial x} + \frac{\partial \sigma_{yy}}{\partial y} + \frac{\partial \sigma_{yz}}{\partial z} = 0 \\ \frac{\partial \sigma_{zx}}{\partial x} + \frac{\partial \sigma_{zy}}{\partial y} + \frac{\partial \sigma_{zz}}{\partial z} = 0 \end{cases} \tag{6}$$

здесь σ_{xx}, σ_{xy}, σ_{yy}, σ_{yz}, σ_{xz}, σ_{zz} – компоненты симметричного тензора напряжений Коши.

Соотношения Коши, связывающие компоненты тензора деформаций $\boldsymbol{\varepsilon}$ с компонентами вектора перемещений $\boldsymbol{u}(u, v, w)$ имеют вид

$$\begin{gathered} \varepsilon_{xx} = \frac{du}{dx}, \quad \varepsilon_{yy} = \frac{dv}{dy}, \quad \varepsilon_{zz} = \frac{dw}{dz} \\ \varepsilon_{yz} = \frac{1}{2}\left(\frac{dw}{dy} + \frac{dv}{dz}\right) \\ \varepsilon_{xz} = \frac{1}{2}\left(\frac{dw}{dx} + \frac{du}{dz}\right) \\ \varepsilon_{xy} = \frac{1}{2}\left(\frac{du}{dy} + \frac{dv}{dx}\right) \end{gathered} \tag{7}$$

Закон Гука имеет вид

$$\begin{gathered} \sigma_{xx} = \lambda(\varepsilon_{xx} + \varepsilon_{yy} + \varepsilon_{zz}) + 2\mu\varepsilon_{xx} ; \\ \sigma_{yy} = \lambda(\varepsilon_{xx} + \varepsilon_{yy} + \varepsilon_{zz}) + 2\mu\varepsilon_{yy} ; \\ \sigma_{zz} = \lambda(\varepsilon_{xx} + \varepsilon_{yy} + \varepsilon_{zz}) + 2\mu\varepsilon_{zz} ; \\ \sigma_{xy} = 2\mu\varepsilon_{xy} , \quad \sigma_{xz} = 2\mu\varepsilon_{xz} , \quad \sigma_{yz} = 2\mu\varepsilon_{yz} ; \end{gathered} \tag{8}$$

В (8) λ и μ - коэффициенты Ляме, которые можно выразить через модуль

Юнга E и коэффициент Пуассона ν следующими соотношениями

$$\lambda = \frac{\nu E}{(1+\nu)(1-2\nu)}$$
$$\mu = \frac{E}{2(1+\nu)} \tag{9}$$

Силовые граничные условия на общем виде выражаются формулами

$$\begin{aligned} S_\sigma :\ & n_x\sigma_{xx} + n_y\sigma_{xy} + n_z\sigma_{xz} = q_{x*} \\ & n_x\sigma_{xy} + n_y\sigma_{yy} + n_z\sigma_{yz} = q_{y*} \\ & n_x\sigma_{xz} + n_y\sigma_{yz} + n_z\sigma_{zz} = q_{y*} \end{aligned} \tag{10}$$

В (10) n_x , n_y , n_z– компоненты внешнего вектора единичной нормали к поверхности, $\mathbf{p}_*$ – заданный вектор внешних сил на границе S_σ.

Граничные условия для перемещений имеют следующий вид:

$$\begin{aligned} S_u :\ & u = u_{1*} \\ & v = u_{2*} \\ & w = u_{3*} \end{aligned} \tag{11}$$

В (11) n_x, n_y , n_z– компоненты внешнего вектора единичной нормали к поверхности, $\mathbf{u}_*$ – заданный вектор перемещений на границе S_u.
Для расчетной схемы, представленной на рис. 4 граничные условия (10) и (11) примут вид

$$\begin{aligned} v|_{\delta_1} &= 0 \\ u|_{\delta_2} &= 0 \\ v|_{\delta_2} &= 0 \\ n\sigma &= q \end{aligned} \tag{12}$$



Рис. 4. Расчётная схема чистого изгиба бруса в теории упругости.

Моделирование чистого изгиба бруса в конечно-элементном пакете FlexPDE

Возможности конечно-элементного пакета FlexPDE позволяют использовать этот пакет для получения численного решения двумерных и трехмерных задач линейной теории упругости и в некоторых случаях удается получить решения при учете физической и геометрической нелинейностях. Опишем скрипт, позволяющий найти решение двумерной задачи чистого изгиба.

В разделе SELECT задаем точность вычислений

```
SELECT
    errlim=1e-5
```

Раздел VARIABLES служит для задания переменных, входящих в дифференциальные уравнения, которые и описывают задачу. При решении двумерной задачи неизвестными, подлежащими определению, являются компоненты вектора перемещений

```
VARIABLES
    U   V
```

В разделе DEFINITIONS задаются все необходимые параметры задачи, к ним можно отнести геометрические и физические характеристики

```
DEFINITIONS
  Ly=1{Ширина балки}
  a=Ly
```

```
  delta=a/10{Ширина заделки}
  eps=a/10{Ширина отрезка нагружения}
  k=2{Коэффициент эффективной длины между заделками}
  m=0.6{}
  Lx=k*Ly+2*m+2*a+2*delta{ Полная длина балки}
  nu=0.3{Коэффициент Пуассона}
  E=200e9{Модуль Юнга}
  Lambda=E/(1+nu)/(1-2*nu) {Коэффициенты Ляме}
  mu=E/(2*(1+nu))
{Деформации}
  ex=dx(U)   ey=dy(V)   exy= dx(V)+ dy(U)
{Напряжения}
  sx=Lambda*((1-nu)*ex+ nu*ey)
  sy=Lambda*(nu*ex+ (1-nu)*ey)
  sxy=mu*exy
  p=0.5e11{распределенная нагрузка}
```

В разделе EQUATIONS задаются уравнения равновесия задачи

```
EQUATIONS
   U:   dx(sx) + dy(sxy)=0
   V:   dx(sxy) + dy(sy)=0
```

В раздели BOUNDARIES описывается геометрия балки и граничные условия

```
BOUNDARIES
     Region 1 "Beam"
     start (0,-Ly/2)
       load(U)=0
       load(V)=0

     {жёсткое закрепление}
```

```
 line to (Lx/2-delta-a,-Ly/2)
 value(U)=x
  value(V) = y

 line to (Lx/2-a,-Ly/2)
  load(U)=0
  load(V)=0
 line to (Lx/2,-Ly/2)
  load(U)=0
  load(V)=0
 line to (Lx/2,Ly/2)
  load(U)=0
  load(V)=-p
line to (Lx/2-eps,Ly/2)
  load(U)=0
  load(V)=0
 line to (-Lx/2+eps,Ly/2)
  load(U)=0
  load(V)=-p
 line to (-Lx/2,Ly/2)
  load(U)=0
  load(V)=0
line to (-Lx/2,-Ly/2)
  load(U)=0
  load(V)=0
 line to (-Lx/2+a,-Ly/2)
  load(U)=0
  value(V)=y{закрепление по оси y}
 line to (-Lx/2+a+delta,-Ly/2)
  load(U)=0
  load(V)=0
line to close
```
Раздел вывода результатов

```
PLOTS
  grid(x, y){недеформированная форма}
  grid( x+U, y+V){деформированная форма}
  contour( sx) painted as "sx" on   grid( x+U, y+V)
  contour( sy) painted as "sy"
END
```

Отличие кода программы для 3D задачи состоит в следующем – в разделе EXTRUSION задается экструзия или растяжение плоской фигуры вдоль оси OZ от начальной нижней плоскости, заданной пользователем, до верхней плоскости.

В разделе COORDINATES для 3D задачи обязательно указываем систему координат

```
COORDINATES    cartesian3
```

Все уравнения и граничные условия задаются для трёхмерной области.



Рис. 5. Деформированная форма. Чистый изгиб бруса. 2D задача.



Рис. 6. Деформированная форма. Чистый изгиб бруса. 3D задача.

Задания

1. Написать скрипты программ для задачи о чистом изгибе бруса в двумерного и трехмерного случая.
2. Сравнить картину напряженно-деформированного состояния от геометрических параметров модели.
3. Построить график зависимости перемещений точки А (центр бруса) от величины нагрузки.
4. Составить отчет о проделанной работе.